# Tarski's definition of finiteness

Johan G. F. Belinfante
2008 August 7

```
In[1]:= SetDirectory["l:"]; << goedel.08aug06a;<< tools.m

        :Package Title: goedel.08aug06a                2008 August 6 at 2:10 p.m.

        It is now:  2008 Aug 7 at 15:53

        Loading Simplification Rules

        TOOLS.M                                    Revised 2008 July 5

        weightlimit = 40
```

## summary

In the **GOEDEL** program, the class **FINITE** of a finite sets was defined by a subvariance condition resembling that used to define the remarkable class **REGULAR**:

```
In[2]:= complement[U[subvar[PS]]]

Out[2]= FINITE

In[3]:= complement[U[subvar[E]]]

Out[3]= REGULAR
```

This analogy made it possible for the author to use William McCune's program **Otter** to find automated proofs of theorems about finiteness by analogy with corresponding theorems about regularity.

```
In[4]:=  "Johan G. F. Belinfante, The unifying concept of subvariance, FTP
          2000, Third International Workshop on First-Order Theorem Proving, St.
          Andrews, Scotland, July, 2000, Fachberichte Informatik, Universität
          Koblenz-Landau, edited by Peter Baumgartner and Hantao Zhang, pp. 56-67.";
```

Among the many theorems about finiteness proved using **Otter** was that this definition of finiteness is equivalent to the statement that a set is finite if and only if it is equipollent to a natural number. This equivalence is the content of the following rewrite rule in the **GOEDEL** program. (Comment: The set of natural numbers is denoted by **omega** and **Q** denotes the equipollence relation.)

```
In[5]:= image[Q, omega]

Out[5]= FINITE
```

A related but somewhat different definition of finiteness was proposed in 1924 by Alfred Tarski.

```
In[6]:= "Alfred Tarski, On finite sets, Fundamenta Mathematicae, vol. 6(1924), pp. 45–95.";
```

Tarski's definition of a finite set is also discussed on pages 99-108 in the following reference:

```
In[7]:= "Patrick Suppes, Axiomatic Set Theory, Dover Publications, New York, 1972.";
```

Tarski defines a set **x** to be "finite" if the restriction of the proper subset relation **PS** to the power set of **x** is well-founded. To avoid any possible confusion, this condition will here be called "Tarski-finiteness."

```
In[8]:= WELLFOUNDED[restrict[PS, P[x], P[x]]]

Out[8]= WELLFOUNDED[composite[id[P[x]], PS]]
```

Any finite set is Tarski-finite:

```
In[9]:= implies[member[x, FINITE], WELLFOUNDED[composite[id[P[x]], PS]]]

Out[9]= True
```

In this notebook it is shown that a class is Tarski-finite if and only if all its subsets are finite. If the axiom of regularity holds, this condition is equivalent to finiteness. Even if the axiom of regularity is not assumed, then it is still true that for the special case of sets Tarski-finiteness and finiteness are equivalent.

## a similar result

Theorem. The following result was previously derived using the **GOEDEL** program in 2006 March 2.

```
In[10]:= SubstTest[equal, set[0], intersection[P[P[x]], subvar[t]], t -> inverse[PS]]

Out[10]= WELLFOUNDED[composite[id[P[x]], inverse[PS]]] == subclass[P[x], FINITE]

In[11]:= WELLFOUNDED[composite[id[P[x_]], inverse[PS]]] := subclass[P[x], FINITE]
```

The key idea used in the derivation of the above result is that if **x** is not a finite set, then the class of finite subsets of **x** is subvariant under the inverse of the proper subset relation.

```
In[12]:= subvariant[inverse[PS], intersection[FINITE, P[x]]]

Out[12]= not[member[x, FINITE]]
```

In this notebook a similar result is derived for **PS**. For this one needs to use cofinite subsets. In the next section it is shown that subvariance under **PS** can be converted to subvariance under **inverse[PS]** via the use of relative complements. The only catch here is that the relative complement function **RC[x]** is empty when **x** is a proper class. So first a temporary theorem is derived for the special case of sets, and later this restriction is removed.

# lemmas about relative complementation

The relative complement function **RC[x]** is defined by the membership rule:

```
In[13]:= member[pair[u, v], RC[x]]

Out[13]= and[equal[0, intersection[u, v]],
          equal[x, union[u, v]], member[u, V], member[v, V], member[x, V]]
```

Here **u** and **v** are relative complements of each other in the set **x**. If **x** is not a set, then **RC[x]** is the empty function.

```
In[14]:= empty[RC[x]]

Out[14]= not[member[x, V]]
```

Lemma.

```
In[15]:= composite[RC[x], Di, RC[x]] // VSRenormality

Out[15]= composite[RC[x], Di, RC[x]] ==
          composite[id[intersection[image[V, set[x]], P[x]]], Di, id[P[x]]]

In[16]:= composite[RC[x_], Di, RC[x_]] :=
          composite[id[intersection[image[V, set[x]], P[x]]], Di, id[P[x]]]
```

Theorem. Converting **inverse[PS]** to **PS** via relative complementation.

```
In[17]:= Map[composite[RC[x], #] &, SubstTest[intersection,
            composite[RC[x], u, RC[x]], composite[RC[x], v, RC[x]], {u → Di, v → S}]] // Reverse

Out[17]= composite[RC[x], inverse[PS], id[P[x]]] == composite[id[P[x]], PS, RC[x]]

In[18]:= composite[RC[x_], inverse[PS], id[P[x_]]] := composite[id[P[x]], PS, RC[x]]
```

Lemma.

```
In[19]:= SubstTest[implies, subclass[u, v],
           subclass[image[t, u], image[t, v]], {t → RC[x], u → intersection[P[x], FINITE],
            v → image[inverse[PS], intersection[P[x], FINITE]]}] // Reverse

Out[19]= or[member[x, FINITE], subclass[image[RC[x], FINITE],
            image[RC[x], image[inverse[PS], intersection[FINITE, P[x]]]]]] == True

In[20]:= (% /. x → x_) /. Equal → SetDelayed
```

Theorem. If **x** is not finite, then the set of all cofinite subsets is subvariant under **PS**.

```
In[21]:= Map[implies[not[member[x, FINITE]],
            subclass[image[RC[x], intersection[FINITE, P[x]]], #]] &,
          ImageComp[composite[RC[x], inverse[PS]], id[P[x]], FINITE]]

Out[21]= or[member[x, FINITE],
           subclass[image[RC[x], FINITE], image[PS, image[RC[x], FINITE]]]] == True

In[22]:= or[member[x_, FINITE],
           subclass[image[RC[x_], FINITE], image[PS, image[RC[x_], FINITE]]]] := True
```

## Tarski finiteness

Lemma.

```
In[23]:= SubstTest[implies, and[member[u, v], subclass[v, w]], member[u, w],
           {u → image[RC[x], FINITE], v -> intersection[P[P[x]], subvar[PS]], w → set[0]}] //
          Reverse

Out[23]= or[not[member[x, V]],
           not[subclass[image[RC[x], FINITE], image[PS, image[RC[x], FINITE]]]],
           not[WELLFOUNDED[composite[id[P[x]], PS]]]] == True

In[24]:= (% /. x → x_) /. Equal → SetDelayed
```

Theorem.  Any Tarski-finite set is finite.  (Comment.  This is equivalent to Theorem **FIN-PP2** proved 2000 March 17 using **Otter**.)

```
In[25]:= Map[not, SubstTest[and, implies[and[p1, p2], p3],
            implies[p3, p4], not[implies[and[p1, p2], p4]],
            {p1 → member[x, V], p2 → WELLFOUNDED[composite[id[P[x]], PS]],
             p3 -> not[subclass[image[RC[x], FINITE], image[PS, image[RC[x], FINITE]]]],
             p4 → member[x, FINITE]}]] // Reverse

Out[25]= or[member[x, FINITE], not[member[x, V]],
           not[WELLFOUNDED[composite[id[P[x]], PS]]]] == True

In[26]:= (% /. x → x_) /. Equal → SetDelayed
```

To remove the sethood limitation, this special case is applied to any infinite subset.  If a class **x** has an infinite subset **y**, then the cofinite subsets of **y** are subvariant under **PS**.  This yields the following result with two variables.  Only the variable **y** here is required to be a set.

```
In[27]:= Map[not, SubstTest[and, implies[and[p1, p2], p3],
            implies[and[p2, p3], p4], not[implies[and[p1, p2], p4]],
            {p1 → WELLFOUNDED[composite[id[P[x]], PS]], p2 → member[y, P[x]],
             p3 -> WELLFOUNDED[composite[id[P[y]], PS]], p4 → member[y, FINITE]}]] // Reverse

Out[27]= or[member[y, FINITE], not[member[y, V]],
           not[subclass[y, x]], not[WELLFOUNDED[composite[id[P[x]], PS]]]] == True

In[28]:= (% /. {x → x_, y → y_}) /. Equal → SetDelayed
```

Theorem. (The set variable **y** is eliminated from the preceding statement.)

```
In[29]:= Map[equal[V, #] &, SubstTest[class, y,
           or[member[y, FINITE], not[member[y, V]], not[subclass[y, x]], not[equal[t, set[0]]]],
           t → intersection[subvar[PS], P[P[x]]]]]

Out[29]= or[not[WELLFOUNDED[composite[id[P[x]], PS]]], subclass[P[x], FINITE]] == True

In[30]:= (% /. x → x_) /. Equal → SetDelayed
```

Main theorem. A class is Tarski-finite if and only if all its subsets are finite.

```
In[31]:= equiv[WELLFOUNDED[composite[id[P[x]], PS]], subclass[P[x], FINITE]]

Out[31]= True

In[32]:= WELLFOUNDED[composite[id[P[x_]], PS]] := subclass[P[x], FINITE]
```

---

## comments

A set is Tarski-finite if and only if it is finite. A rewrite rule for this can be introduced.

```
In[33]:= equiv[and[member[x, V], subclass[P[x], FINITE]], member[x, FINITE]]

Out[33]= True

In[34]:= and[member[x_, V], subclass[P[x_], FINITE]] := member[x, FINITE]
```

If the axiom of regularity holds, any Tarski-finite class is a finite set.

```
In[35]:= SubstTest[implies, and[equal[v, REGULAR], subclass[P[x], intersection[FINITE, v]]],
           member[x, FINITE], v → V] // Reverse

Out[35]= or[member[x, FINITE], not[AxReg], not[subclass[P[x], FINITE]]] == True

In[36]:= or[member[x_, FINITE], not[AxReg], not[subclass[P[x_], FINITE]]] := True
```

Restatement: Finiteness and Tarski-finiteness are equivalent if **AxReg** holds.

```
In[37]:= implies[AxReg, equiv[subclass[P[x], FINITE], member[x, FINITE]]]

Out[37]= True
```

An example. The class **OMEGA** of ordinals is well-founded with respect to the proper subset relation, but its power class is not.

```
In[38]:= WELLFOUNDED[composite[id[OMEGA], PS]]

Out[38]= True
```

```
In[39]:= WELLFOUNDED[composite[id[P[OMEGA]], PS]]

Out[39]= False
```

Intuitively, there can be no infinitely descending chain of ordinal numbers $\cdots < n_2 < n_1 < n_0$. (Here " < " denotes proper inclusion.) One can however have an infinitely descending chain of sets of ordinal numbers, because from any infinite set of ordinals one can keep on removing more and more elements forever.